

Darstellung von Stromlinien mittels komplex-analytischer, konformer (meromorpher) Abbildungen

Dipl.-Ing.(FH) Kapt.(AG) Wolf Scheuermann

Hamburg, 2009

Contents

1	Abkürzungen	1
2	Potentialströmung	2
2.1	Strömung um rotierenden Zylinder	2
2.2	Berechnung der Stromlinie	3
3	Computer-Programm	4
4	Stromlinienbilder von Profilen	6
4.1	Aerodynamik mit meromorphen Funktionen	6
4.2	Aerodynamik nach numerischer Panel-Methode	7
4.3	Vergleich der Strombilder	9
5	Schluß	11
6	Quellen	11

1 Abkürzungen

α : Polarkoordinate: Winkel
 r : Polarkoordinate: Abstand
 Γ : Zirkulation
 M : Dipolmoment
 t : Profiltiefe
 d : Zylinderradius
 c : Wölbung (Camber)
 u : Anströmgeschwindigkeit
 π : Kreiszahl

2 Potentialströmung

Potentialströmungen werden durch komplexwertige, analytische Funktionen

$$W = \varphi + i\psi$$

gebildet, welche folgendes Kriterium erfüllen [1]:

$$W' = \frac{\partial\varphi}{\partial x} + i\frac{\partial\psi}{\partial x} = -i\frac{\partial\varphi}{\partial y} + \frac{\partial\psi}{\partial y} \quad (1)$$

Solche komplex-differenzierbaren, holomorphen Funktionen sind immer auch konform, also winkeltreu. Für die Abbildungen des komplexen Einheitskreises werden i.a. meromorphe Funktionen als Verallgemeinerung holomorpher Funktionen verwendet. Beispiele sind die Joukowski-Transformation oder die v.Mises-Profil-Formel [2].

Die Lösungsfunktion φ des Gleichungssystems (1) ist dabei die Potentialfunktion und ψ die Stromfunktion. Solche holomorphen Funktionen W sind addierbar und beschreiben ihrerseits wiederum Potentialströmungen. Beispiel hierfür sind die Transformationen, welche den Einheitskreis in die Profile der Sc-Serie überführen:

z : Kreis zentriert um m_1 mit Radius 1

v.Mises "Zwischen-Profil":

$$z_1 = z + \frac{b_1}{z} + \frac{b_2}{z^2}$$

Theodorsen Profil:

$$W(z) = z_2 = z_1 + \frac{a_1}{z_1} + \frac{a_2}{z_1^2} + \frac{a_3}{z_1^3}$$

Die Real- und Imaginärteile der Koeffizienten a_i , b_j und m_1 bestimmen die Form des Profils und obendrein das konforme Bild eines jeden Punktes der komplexen Ebene.

2.1 Strömung um rotierenden Zylinder

Die Umströmung eines Zylinders kann durch Addition der parallelen Anströmung mit einer Dipolströmung modelliert werden. Addiert man dazu das Strömungsfeld eines Potentialwirbels, so ergibt sich die Strömung um einen rotierenden Zylinder [3].

Im Folgenden werden die Gleichungen elementarer Potentialströmungen aufgelistet.

Die Funktionswerte sind hierbei Funktionen von Polarkoordinaten: r , α

In kartesischen Koordinaten (x,y) stellt x den realen Achsabschnitt und y den imaginären Achsabschnitt des Punktes in der komplexen Ebene dar.

Parallelströmung

Eine Parallelströmung wird dargestellt durch:

$$\psi(r, \alpha) = u \cdot r \cdot \sin(\alpha)$$

Dipol

Eine Dipolströmung (Quelle und Senke in einem Punkt) wird beschrieben vermittels:

$$\psi(r, \alpha) = -\frac{M}{2\pi \cdot r} \cdot \sin(\alpha)$$

Wobei M das Dipolmoment ist:

$$M = 2\pi \cdot u \cdot d^2$$

Potentialwirbel

Ein Potentialwirbel hat die Gleichung:

$$\psi(r, \alpha) = \frac{\Gamma}{2\pi} \cdot \ln(r)$$

Potentialströmung um rotierenden Zylinder

Parallelströmung + Dipol + Potentialwirbel = umströmter, rotierender Zylinder:

$$\psi(r, \alpha) = u \cdot r \cdot \sin(\alpha) \cdot \left(1 - \frac{d^2}{r^2}\right) + \frac{\Gamma}{2\pi} \ln(r) \quad (2)$$

2.2 Berechnung der Stromlinie

Mit Gleichung (2) lassen sich Stromlinien im Falle des umströmten, rotierenden Zylinders geschlossen lösen.

Die folgende Umformung...

$$\sin(\alpha) = \frac{\psi - \frac{\Gamma \cdot \ln(r)}{2\pi}}{u \cdot r \cdot \left(1 - \frac{d^2}{r^2}\right)} \quad (3)$$

...erlaubt die Auflösung nach α wenn die Zirkulation Γ festgelegt, der Wert ψ der Stromlinie gegeben und ein bestimmter Abstand r vom Zylinderkreis mit Radius d gewählt ist.

Der Punkt (r, α) ist somit ein Punkt der Stromlinie, der mittels der Transformation

$$\begin{aligned} x &= r \cdot \cos(\alpha) \\ y &= r \cdot \sin(\alpha) \end{aligned} \tag{4}$$

in die kartesische Ebene transformiert werden kann. Lassen wir r mit d beginnend aufwärts laufen, so erhalten wir, ggf. unter Berücksichtigung der Quadranten, den Verlauf der Stromlinie ψ .

Ein einfaches Programm, das diese Aufgabe löst, ist im nächsten Abschnitt gegeben.

3 Computer-Programm

Hier ist ein Programm in Microsoft *QBasic*TM wiedergegeben, das Stromlinien eines rotierenden Zylinders punktweise findet. Werden diese Punkte konform in die komplexe Ebene abgebildet, mittels der Funktionen die den Kreis zum Profil transformieren, so ergeben sich punktweise die Stromlinien des umströmten Profils.

```
'STREAMLINES

DECLARE FUNCTION asin! (x AS SINGLE) 'Arcus Sinus

SCREEN 12 'Graphic Display
VIEW (0, 0)-(479, 479), 15
WINDOW (-100, 100)-(100, -100)

PI = ATN(1) * 4 'Ludolff's Number
RAD = PI / 180 'factor deg -> rad

stepp = .3      'radial step
g = 130        'circulation
d = 20         'radius of rotating cylinder
u = 1          'free stream speed
psi0 = 62      'stream function value at stagnation line
'(r,alfa)     'polar coordinate of stream line with value psi = const
'(x,y)        'cartesic coordinate of stream line with value psi = const

CIRCLE (0, 0), d, 12 'draw the cylinder
FOR r = d + stepp TO 7.5 * d STEP stepp 'step through the radial range
  FOR psi = psi0 TO 250 STEP 10 'streamline at one side
    'calculate the angle where at the given range psi is found
    sinalfa = (psi - g * LOG(r) / (2 * PI)) / (u * r * (1 - d*d / (r*r)))
```

```

IF ABS(sinalfa) > 1 THEN EXIT FOR 'avoid numerical errors
alfa = asin(sinalfa)
x = r * COS(alfa) 'transform into cartesian coordinates
y = r * sinalfa
PSET (x, y), 0 'plot the streamline
PSET (-x, y), 0
NEXT
FOR psi = psi0 TO -100 STEP -10 'streamline at the other side
'calculate the angle where at the given range psi is found
sinalfa = (psi - g * LOG(r) / (2 * PI)) / (u * r * (1 - d*d / (r*r)))
IF ABS(sinalfa) > 1 THEN EXIT FOR 'avoid numerical errors
alfa = asin(sinalfa)
x = r * COS(alfa) 'transform into cartesian coordinates
y = r * sinalfa
PSET (x, y), 0 'plot the streamline
PSET (-x, y), 0
NEXT
NEXT
CIRCLE (0, 0), d, 12 'draw the cylinder again
END 'End of Program

FUNCTION asin (x AS SINGLE) 'Arcus Sinus Function
  asin = ATN(x / SQR(1 - x * x))
END FUNCTION 'End of Function

```

Die Ausgabe dieses Programms sieht folgendermaßen aus:




Figure 1: Stromlinien um einen angeströmten, rotierenden Zylinder

4 Stromlinienbilder von Profilen

Nachfolgend wird das Stromlinienbild eines Profils gemäß dem oben beschriebenen Verfahren erzeugt. Mit Hilfe eines aerodynamischen Panel-Verfahrens wird das Stromlinienbild rekonstruiert und (manuell) zur Prüfung der Übereinstimmung zur Deckung gebracht.

4.1 Aerodynamik mit meromorphen Funktionen

Ausgangspunkt ist ein selbstgeschriebenes *VisualBasicTM*-Programm zur Erzeugung von Profilen der Sc-Serie mittels meromorpher Funktionen. Hier ist das zu transformierende Stromlinienbild eines angeströmten, rotierenden Zylinders gezeigt:




Figure 2: Angeströmter, rotierender Zylinder

Dieser Zylinder und die Stromlinien werden anschließend durch geeignete Wahl der komplexen Parameter in ein Profil transformiert:




Figure 3: Umströmtes Profil

4.2 Aerodynamik nach numerischer Panel-Methode

Die Stromlinien dieses Profils wurden zusätzlich auf gänzlich andere Art, mit Hilfe numerischer Panel-Methoden berechnet. Zur Anwendung kam das Programm ViewProf 1.02 [4].

ViewProf ist ein Profil-Darstellungsprogramm, das eine Multiblock 2D-Lösung für ideale Strömungen nach Oellers Methode bietet.




Figure 4: Numerische Panel-Methode

4.3 Vergleich der Strombilder

Manuell und nach Augenmaß wurden die beiden Strömungsbilder ausgeschnitten, gedreht und normiert und die Strömungsparameter angepaßt, um sie vergleichbar zu machen:




Figure 5: Strömungsbild nach Panel-Methode




Figure 6: Strömungsbild nach komplex-analytischer Methode

Anschließend wurden beide Bilder übereinander gelegt, um die Übereinstimmung prüfen zu können:




Figure 7: Vergleich der Strömungsbilder

Man kann von guter Übereinstimmung der beiden Lösungen sprechen, wenn man die Ungenauigkeiten infolge der manuellen Anpassung außer Acht läßt.

Ich wage sogar die Vermutung, daß beide Methoden äquivalent sind oder die komplex-analytische Methode aufgrund der geschlossenen, nichtiterativen Lösung die genauere von beiden ist.

5 Schluß

Die hier vorgestellte komplex-analytische Methode zur Darstellung von Stromlinien angeströmter Profile scheint herkömmlichen numerischen Methoden gleichwertig oder aufgrund der geschlossenen Lösung sogar überlegen zu sein.

Wegen der Nähe zu den Potentialtheoretischen Funktionen lassen sich weitere Darstellungen ableiten wie Geschwindigkeitsfeld oder Druckverteilung.

Eventuell sind sogar Erweiterung der Potentialtheorie, wie z.B. Ergänzung eines Terms zur Modellierung der Grenzschicht möglich, welche die Darstellung von jetzt ca. 80% Realitätsnähe steigert.

6 Quellen

References

- [1] Jack Moran:
An Introduction to Theoretical and Computational Aerodynamics.
Dover Publications Inc.,
Mineola, New York 1984
- [2] Richard von Mises:
Theory of Flight.
Dover Publications Inc.,
Cambridge, Mass. 1945
- [3] Ernst Käppeli:
Aufgabensammlung zur Fluidmechanik.
Teil 1, Potentialströmungen.
Verlag Harri Deutsch,
Thun, Frankfurt am Main 1992
- [4] Matti Hyötyniemi:
ViewProf 1.02, public domain software.
Finland, Järvenpää 1995